Equilibrium of Bodies

5.1E Three-Dimensional Plots

EES provides the capability to plot a dependent variable as a function of two independent variables in what are referred to as three-dimensional (or X-Y-Z) plots. EES can generate different types of 3-D plots including the two that are covered in this section: contour plots and 3-D surface plots. We will discuss these plots in the context of a simple example function

$$z = 4\exp\left(-\frac{y^2}{4}\right)\sin\left(2\,x - 1\left[\text{rad}\right]\right). \tag{5.1}$$

This function can be entered in EES as shown below.

```
$UnitSystem Radian

x = 0 [rad]
y = 1
z = 4*Exp(-y^2/4)*Sin(2*x - 1 [rad])
```

Solving these equations results in z = -2.621.

Three Column Data

All of the 3-D plot types require three-dimensional data. These data can either be provided in three column form or in the form of a 2-D table. In three column form, the data are provided as three separate columns corresponding to the two independent variables and the dependent variable. For this problem we can provide the data in three column form by creating a Parametric Table that includes the variables x, y, and z.

In three column form, the independent variables can be provided in any order and they do not have to be located on a regular grid. The first step that EES will do internally is interpolate/extrapolate the provided data to provide a suitable grid. To generate the data that will be plotted, we will vary the independent variable x from -3 rad to 3 rad in 41 discrete values. The dependent variable y will be varied from -5 to 5, also in 41 discrete values. Therefore, the Parametric Table must contain 41x41 = 1681 rows. To populate the column containing x, right click on the column header and select Alter Values to obtain the dialog shown in Figure 5.1(a). Vary x from -3 rad to 3 rad and select Repeat pattern every 41 rows, as shown.



(a) (b)

Figure 5.1. Alter values dialog for (a) *x* and (b) *y*.

To populate the column containing *y*, repeat the same process but vary *y* from -5 to 5 and select Apply pattern every 41 rows, as shown in Figure 5.1(b). If you examine your Parametric Table you will find that the dependent variables form a 2-D grid over the computational domain -3 rad $< x <$ 3 rad and -5 $< y <$ 5. Comment out the values of *x* and *y* that were previously entered in the Equations Window:

```
{x = 0 [rad]
y = 1}
```

and solve the table to obtain a set of data in three column format.

Contour-Lines Plot

Select New Plot Window from the Plots menu and then select X-Y-Z Plot to access the X-Y-Z Plot Setup dialog shown in Figure 5.2. A contour plot illustrates lines of constant values of the dependent variable (i.e., isometric lines) in the parameter space defined by the two independent variables. In order to generate a contour plot, select Contour-Lines from the list of plot types.



Figure 5.2. X-Y-Z Plot Setup Dialog for a Contour-Lines plot.

Select variables x and y as the independent variables and z as the contour variable. Adjust the minimum, maximum and interval of each variable. The data in the columns must be interpolated/extrapolated by the internal plotting algorithms to provide values of the independent variable over the entire selected range of the dependent variables at relatively fine intervals. This process will be done using either Radial basis functions or Bi-quadratic polynomials depending on the selection from the drop-down menu in the lower right corner of the dialog. The grid size is controlled by the resolution slider. The smoothing slider is useful for experimental data or for a very sparse data set where outliers or large gaps may be present; normally the smoothing slider should be set to zero. Select OK in order to generate the contour plot shown in Figure 5.3.



Figure 5.3. Contour-lines plot showing z as a function of x and y.

Contour-Bands Plot

The Contour-Bands option in the X-Y-Z Plot Setup dialog shown in Figure 5.2 produces the plot shown in Figure 5.4. Each contour band is indicated by a color as indicated in the legend.



Figure 5.4. Color band contour plot using full spectrum color.

Three color schemes are available. The color scheme shown in Figure 5.4 is called the Full Spectrum color scheme and it uses all of the colors of the rainbow. An alternative color scheme uses colors ranging from blue to red and a third option is gray scale. To change the color scheme, click the right mouse button anywhere within the plot rectangle in order to display the Modify Plot Dialog.

3D Surface Plot

The 3D Surface plot type uses data in any EES table along with interpolation and extrapolation procedures to generate a 3D surface plot. Select 3D Surface from the list of plot types in the X-Y-Z Plot Setup dialog (Figure 5.2) in order to display a three-dimensional, rotatable projection of the surface. The resulting plot is shown in Figure 5.5.

A control panel appearing as shown in Figure 5.6 is provided at the bottom of the 3D plot window. If the panel is not visible, click the right mouse button anywhere in the plot and it will appear. The same action will make it disappear. Alternatively, you can use the Show/Hide Tool Bar menu item in the Plots menu to control the visibility of the control panel. The control panel allows you to manipulate various aspects of the plot like the axes, the legend (i.e., the color bar), the position of the grid, the perspective, etc.



Figure 5.5. 3D Surface plot of *z* as a function of *x* and *y*.



Figure 5.6. Control Panel for 3-D Surface Plots.

The **Color** check box has three positions that engage with a mouse click. The three possibilities are black and white (no check); color display with colors ranging from blue to red (gray check); and color display with full spectrum color (black check). Normally, the color (or shade of gray for black and white plots) is a visual indicator of the *z*-axis variable. However, it is possible to have the color correspond to a totally different piece of information, which effectively provides a 4-D plot.

To demonstrate this capability, we will assign the color of the surface plot to another variable c which is set to the distance from the origin in the $z = 0$ plane.

```
c=Sqrt(x^2+y^2)
```

Add a fourth column for the variable c to the Parametric table and re-solve the table. Select 3D Surface from the list in the X-Y-Z Plot Setup dialog in Figure 5.2. Select z to be the Z-axis variable and then click on the **[Z-Axis Variable]** label located above the third list box to select the variable used to specify the color. Notice that the label changes to **[Color Variable]** and the list box will be displayed in yellow. Select variable C, as shown in Figure 5.7, and the click the OK button. The plot shown in Figure 5.8 will be generated.



Figure 5.7. X-Y-Z Plot Setup Dialog for a 3D Surface plot with the color variable selected.



Figure 5.8. 3D Surface plot of z as a function of x and y with color of surface set by the variable c.

EXAMPLE E5.1 *Equilibrium Analysis*

We will revisit Example 5.1 from the text. A device for clamping a flat workpiece in a machine tool is shown.



Figure 1. Problem from Example 5.1 in Gray et al. (2023).

If a 200 N clamping force is to be generated at *C*, and if the contact at *C* is smooth (no friction), determine the force *F* required and the reaction at *B*.

SOLUTION

Road Map The solution proceeds by initially defining the coordinates of the mechanism. The coordinates *D* and *B* define the direction of the reaction at *B*. This leads to three unknowns corresponding to the two components of the force at *A* and the vertical component of the force at *C*. The three equations used to specify these are the equilibrium equations corresponding to sum of forces in *x* and *y* must both be zero and sum of moments about point *A* must be also zero.

Governing Equations The coordinates *A*, *B*, *C*, and *D* are all known. The position vector $\vec{r}_{BD}$ is defined by

$$\vec{r}_{BD} = D - B \tag{1}$$

and defines the direction of the force applied at point *B*, $\vec{F}_{BD}$. A free body diagram on the clamp is shown in Figure 2.



Figure 2. Free body diagram on the clamp.

The force applied at point *C* ($\vec{F}_C$, the clamping force) is completely specified as 200 N in the *y*-direction. The force applied at point *B* is given by

$$\vec{F}_B = F_B \frac{\vec{r}_{BD}}{\left|\vec{r}_{BD}\right|}. \tag{2}$$

The force equilibrium equations can be written as

$$\vec{F}_A + \vec{F}_B + \vec{F}_C = \vec{0}. \tag{3}$$

The moment equilibrium equation requires that the sum of moments about any point on the clamp be zero. In this problem we will take the sum of moments about point A

$$\vec{r}_{AB} \times \vec{F}_B + \vec{r}_{AC} \times \vec{F}_C = 0, \tag{4}$$

where

$$\vec{r}_{AB} = B - A, \text{ and} \tag{5}$$

$$\vec{r}_{AC} = C - A. \tag{6}$$

Equations (3) and (4) are three equations in the three unknowns $F_{A,x}$, $F_{A,y}$, and F_B.

Computation The coordinates A, B, C, and D are defined using a coordinate system with an origin defined as being the x location of point A and the y location of point D.

```
"coordinates"
$Vector2D A, B, C, D
A = VectorAssign(0,32) [mm]
B = VectorAssign(80, 24) [mm]
C = VectorAssign(130,4) [mm]
D = VectorAssign(35,0) [mm]
```

The position vector $\vec{r}_{BD}$ is calculated using Eq. (1). The clamping force, $\vec{F}_C$, is defined as given in the problem statement and the force $\vec{F}_B$ is entered as shown in Eq. (2).

```
$Vector2D r_BD
r_BD = D - B                                   "member BD"

$Vector2D F_A, F_B, F_C
F_C = VectorAssign(0, 200 [N])                 "clamping force"
F_B = magF_BD*r_BD/VectorMag(r_BD)             "force applied at location B"
```

The equilibrium equations given by Eqs. (3) through (6) are entered.

```
F_A + F_B + F_C = VectorZeros                  "force equilibrium equation"

$Vector2D r_AB, r_AC
r_AB = B - A
```

```
r_AC = C - A

VectorCross(r_AB,F_B) + VectorCross(r_AC, F_C) = 0 "moment equil. eq."
```

Solving leads to F_{BD} = 581.6 N and $\vec{F}_A$ = (513.2, 73.68) N.

Discussion & Verification We can visualize the results by making a 2-D vector plot. To make a force polygon we would use the $VectorPlot2D directive shown below

```
$VectorPlot2D Name = 'Force Polygon' F_A:0/red F_B:!/red F_C:!/red
```

which results in the plot shown in Figure 3.



Figure 3. Force polygon for the clamp.

We can also examine the impact of the location of point *B* on the required force applied at point *A* in the *x*-direction ($F_{A,x}$) by creating a 3-D plot as discussed in Section 5.1. Let's comment out the coordinates for point *B*

```
{B = VectorAssign(80, 24) [mm]}
```

and instead set the values of B_x and B_y within a Parametric Table that also includes $F_{A,x}$. We will vary B_x from 40 mm to 125 mm and vary B_y from 0 mm to 32 mm. Once the table has been solved, we can make either a contour or surface plot of the results, both are shown in Figure 4. Note that the smallest force is required when the location of point *B* is as far up and to the left as possible over the range that we investigated.



Figure 4. Required force in the x-direction at point A as a function of the x- and y-coordinates of point B.

EXAMPLE E5.2 *Two-Dimensional Idealization of a Three-Dimensional Problem*

We will revisit Example 5.3 from the text. The rear door of a minivan is hinged at point A and is supported by two struts; one strut is between points B and C, and the second strut is immediately behind this on the opposite side of the door. If the door weighs 350 N with center of gravity at point D and it is desired that a 40 N vertical force applied by a person's hand at point E will begin closing the door, determine the force that each of the two struts must support and the reaction at the hinge.



Figure 1. Problem 5.3 in Gray et al. (2023).

SOLUTION

Road Map The problem is a relatively simple application of the equilibrium equations in two dimensions. We need to require that the sum of forces and sum of moments are zero. Because this is a two-dimensional problem, the sum of forces will provide two equations and the sum of moments only one. Because the direction of the strut force is known we have three unknowns corresponding to its magnitude as well as the reactions at location *A* in both directions.

Governing Equations The geometry is specified by the points *A*, *B*, *C*, *D*, and *E*. The force from the hand (F_{hand}) is vertically down and applied at point *E*

$$\vec{F}_E = -F_{hand}\,\hat{j}. \tag{1}$$

The force associated with the weight of the door (*W*) is vertically down and applied at point *D*, the center of mass

$$\vec{F}_D = -W\,\hat{j}. \tag{2}$$

The direction associated with the force from the struts is defined by the position vector $\vec{r}_{CB}$. Therefore the force applied at location *B* is

$$\vec{F}_B = 2\,F_{strut}\frac{\vec{r}_{CB}}{\left|\vec{r}_{CB}\right|}. \tag{3}$$

where F_{strut} is the magnitude of the force applied by each of the individual struts. The force equilibrium equation is

$$\vec{F}_A + \vec{F}_B + \vec{F}_D + \vec{F}_E = \vec{0}. \tag{4}$$

The moment equilibrium equation is taken about point *A* and leads to

$$\vec{r}_{AB}\times\vec{F}_B + \vec{r}_{AD}\times\vec{F}_D + \vec{r}_{AE}\times\vec{F}_E = 0. \tag{5}$$

Computation The inputs include the coordinates of the points as well as the magnitude of the force from the hand and the weight.

```
$Vector2D A, B, C, D, E                         "assign coordinates for problem"
A = VectorAssign(1150,200) [mm]
B = VectorAssign(500,450) [mm]
C = VectorAssign(1070,0) [mm]
D = VectorAssign(350,550) [mm]
E = VectorAssign(0,450) [mm]

F_hand = 40 [N]                                 "magnitude of force from hand"
```

```
W = 350 [N]                                       "weight of door"
```

The four forces are defined as being 2-D vectors along with the position vector $\vec{r}_{CB}$. Equations (1) through (4) are entered.

```
$Vector2D F_E, F_D, F_B, F_A, r_CB
F_E = -F_hand*VectorUnit_j                        "force from hand"
F_D = -W*VectorUnit_j                             "weight"
r_CB = B - C                                      "position vector from C to B"
F_B = 2*F_strut*r_CB/VectorMag(r_CB)              "force from struts"

F_E + F_D + F_B + F_A = VectorZeros               "sum of forces = 0"
```

Finally the position vectors required by Eq. (5) are defined as being 2-D vectors and computed and then Eq. (5) is entered.

```
$Vector2D r_AB, r_AD, r_AE
r_AB = B - A
r_AD = D - A
r_AE = E - A
VectorCross(r_AB,F_B) + VectorCross(r_AD,F_D) + VectorCross(r_AE,F_E) = 0
                                                  "sum of moments about A must be zero"
```

Solving provides F_{strut} = 789.2 N and $\vec{F}_A$= (1239, -588) N, which agrees with the answer in the text.

Discussion & Verification It is instructive to visualize these results using a 2-D vector plot. The $VectorPlot2D directive below will plot each of the forces involved, locating them at their point of application.

```
$VectorPlot2D Name = 'VectorPlot' F_A:A/red F_B:B/red F_D:D/red F_E:E/red
```

The directive leads to the plot shown in Figure 2. Note that the forces associated with the strut and the reaction force on the hinge are much larger than either the force from the hand or the weight of the door.



Figure 2. 2-D vector plot showing the forces acting on the door.

Finally, we can examine the design of the door using a 3-D plot. Here let's assume that the struts provide a constant force, F_{strut} = 789.2 N, and see how the force required by the hand varies with the position of coordinate B. First we enter the specified strut force and then comment out the hand force and the coordinate of B.

```
F_strut = 789.2 [N]
{F_hand = 40 [N]}                "magnitude of force from hand"
{B = VectorAssign(500,450) [mm]}
```

A Parametric Table is created with 100 rows and the variables B_x, B_y, and F_hand are included. The value of B_x is varied from 400 mm to 700 mm and the value of B_y is varied from 300 mm to 600 mm. The results are used to generate the surface plot shown in Figure 3. Note that as the strut mounting location is moved towards the rear of the car or vertically down the force required by the hand to close the door increases.



Figure 4. Magnitude of force from the hand required to close the door as a function of the coordinates of the mounting position of the strut.